

Лабораторная работа № 5

Статическое деформирование пьезокерамического преобразователя с многоэлектродным покрытием под действием заданной разности потенциалов на электродах

Цель работы:

1. Изучить основы программирования на языке APDL программного комплекса ANSYS на примере решения статической осесимметричной задачи для пьезоэлектрического диска с многоэлектродным покрытием (файл FEM_PD_St1.inp):

- а) задание констант пьезоэлектрического материала;
- б) элементные системы координат и их применение для задания анизотропных и пьезоэлектрических свойств;
- в) использование регулярного разбиения
- г) задание электродированных поверхностей и граничных условий на них.

2. Составить программу на языке APDL ANSYS для индивидуальной статической задачи для пьезоэлектрического диска с многоэлектродным покрытием.

3. Проанализировать результаты и оформить отчет.

Пример

Пьезоэлектрический диск радиуса R и толщины H отнесен к цилиндрической системе координат $O r z$ ($0 \leq r \leq R, -H/2 \leq z \leq H/2$) и находится в условиях осесимметричного деформирования. В силу осесимметричности задачи далее рассматривается только его меридиональное сечение, причем в соответствие с методологией ANSYS ось r считается осью X , а ось z – осью Y .

Диск имеет четыре электродированные поверхности (рис. 1): электрод с номером 1 – $0 \leq X \leq R_1, Y = -H/2$; электрод с номером 2 – $0 \leq X \leq R_1, Y = H/2$; электрод с номером 3 – $R_2 \leq X \leq R, Y = -H/2$; электрод с номером 4 – $R_2 \leq X \leq R, Y = H/2$. Пьезоэлектрический диск выполнен из пьезокерамики PZT-4, причем зоны $\{0 \leq X \leq R_1, -H/2 \leq Y \leq H/2\}$ и $\{R_1 \leq X \leq R_2, -H/2 \leq Y \leq H/2\}$ поляризованы вдоль оси Y , а зона $\{R_2 \leq X \leq R, -H/2 \leq Y \leq H/2\}$ поляризована противоположно оси Y (Рис. 2).




Рис. 1 Схема пьезоэлектрического диска




Рис. 2 Схема меридионального сечения пьезоэлектрического диска

Диск закреплен по оси Y в зоне $X = R$, $Y = 0$, т.е. $U_y = 0$ при $X = R$, $Y = 0$, на оси $X = 0$ следует задать условия симметрии, а остальные поверхности диска свободны от механических напряжений.

Диск деформируется под действием подаваемого электрического напряжения на электроды: $\varphi = -V_{inp}$ – на первом электроде, $\varphi = V_{inp}$ – на втором электроде, $\varphi = -V_{inp}$ – на третьем электроде, и $\varphi = V_{inp}$ – на четвертом электроде.

Необходимо определить смещения диска в результате деформирования.

РЕШЕНИЕ ЗАДАЧИ С ИСПОЛЬЗОВАНИЕМ ANSYS

Пример решения задачи с использованием программы ANSYS содержится в файле FEM_PD_St1.inp. Программа предназначена для выполнения в пакетном режиме ANSYS и снабжена подробными комментариями. Другие примеры решения задач электроупругости содержатся в документации по ANSYS: Coupled-Field Guide, 2.14. Sample Piezoelectric Analysis (Batch or Command Method), 2.16. Sample Electroelastic Analysis of a Dielectric Elastomer (Batch or Command Method); Verification Manual, файлы Vm175.dat, Vm176.dat, Vm231.dat, Vm237.dat (ссылки приведены для ANSYS 11.0).

Начинать работу с программным комплексом ANSYS лучше с запуска ANSYS Product Launcher (Пуск→Все программы), где следует задать среду моделирования (Simulation Environment→ANSYS), выбрать рабочий каталог, в который будут сохраняться все рабочие файлы ANSYS (Working Directory) и задать название проекта (Job Name).

Выполнение *командного файла* (текстовый файл с расширением .inp или .txt), написанного на языке APDL ANSYS осуществляется с помощью пункта меню File → Read Input from... В процессе создания командного файла можно копировать блоки команд в командную строку и смотреть на результат выполнения.

Периодически следует сохранять результаты работы! Перед запуском новых расчетов надо очищать текущую базу данных (File→Clear and Start New).

Определение материальных констант пьезоэлектрических материалов.

Как известно, пьезоэффект наблюдается лишь в кристаллах, не имеющих центральной симметрии. В связи с этим, пьезоэлектрические тела обязательно должны обладать анизотропными свойствами. В общем случае в ANSYS для задания материальных констант пьезоэлектрических тел необходимо определить следующие величины: плотность ρ ; симметричную матрицу упругих модулей $c_{\alpha\beta}^E$; $\alpha, \beta = 1, \dots, 6$; $c_{\alpha\beta}^E = c_{\beta\alpha}^E$; матрицу пьезомодулей $e_{i\alpha}$; $i = 1, 2, 3$; $\alpha = 1, \dots, 6$; и диагональную матрицу диэлектрических проницаемостей ε_{ii}^S ; $i = 1, 2, 3$. (Можно задавать и альтернативные наборы констант типа матрицы упругих податливостей $s_{\alpha\beta}^E$ и др.)

Опишем технику определения в ANSYS модулей $c_{\alpha\beta}^E$ и $e_{i\alpha}$ более подробно. Коэффициенты $c_{\alpha\beta}^E$ задаются в форме 6x6 матрицы (4x4 – для двумерных задач). В силу симметрии матрицы упругих модулей используется только ее верхняя треугольная часть, причем при нестандартном расположении коэффициентов:

$$\mathbf{c}_{\text{ANSYS,3D}} = \begin{array}{c|cccccc} & x & y & z & xy & yz & xz \\ \hline x & c_{11}^E & c_{12}^E & c_{13}^E & c_{16}^E & c_{14}^E & c_{15}^E \\ y & & c_{22}^E & c_{23}^E & c_{26}^E & c_{24}^E & c_{25}^E \\ z & & & c_{33}^E & c_{36}^E & c_{34}^E & c_{35}^E \\ xy & & & & c_{66}^E & c_{46}^E & c_{56}^E \\ yz & & \text{sym} & & & c_{44}^E & c_{45}^E \\ xz & & & & & & c_{55}^E \end{array} \quad (1.1)$$

$$\mathbf{c}_{\text{ANSYS,2D}} = \begin{array}{c|cccc} & x & y & z & xy \\ \hline x & c_{11}^E & c_{12}^E & c_{13}^E & c_{16}^E \\ y & & c_{22}^E & c_{23}^E & c_{26}^E \\ z & & & c_{33}^E & c_{36}^E \\ xy & & \text{sym} & & c_{66}^E \end{array} \quad (1.2)$$

где $c_{\alpha\beta}^E$ – общепринятые в теории пьезоэлектричества обозначения для упругих модулей.

Матрицы $c_{ANSYS,3(2)D}$ в ANSYS задаются обычно командами: `TB,ANISO,MAT` и `TBDATA,STLOC,C1,C2,...,C6`; где *MAT* – номер набора материальных свойств, *STLOC* – номер для начала последовательного размещения данных *C1,C2,...,C6* в одномерном массиве данных. Массив данных для коэффициентов, входящих в (1.1), (1.2), заполняется по строкам (1.1) в виде одномерного массива с 21 компонентой:

$$\begin{bmatrix} 1 & 2 & 3 & 4 & 5 & 6 \\ & 7 & 8 & 9 & 10 & 11 \\ & & 12 & 13 & 14 & 15 \\ & & & 16 & 17 & 18 \\ & & & & 19 & 20 \\ & & & & & 21 \end{bmatrix} \quad (1.3)$$

В результате получается следующее соответствие между данными из массива, определенного `TBDATA`, и модулями упругости $c_{\alpha\beta}^E$:

№	1	2	3	4	5	6	7	8	9	10	
$c_{\alpha\beta}^E$	c_{11}^E	c_{12}^E	c_{13}^E	c_{16}^E	c_{14}^E	c_{15}^E	c_{22}^E	c_{23}^E	c_{26}^E	c_{24}^E	
№	11	12	13	14	15	16	17	18	19	20	21
	c_{25}^E	c_{33}^E	c_{36}^E	c_{34}^E	c_{35}^E	c_{66}^E	c_{46}^E	c_{56}^E	c_{44}^E	c_{45}^E	c_{55}^E

Для наиболее распространенного типа пьезоэлектрических материалов – пьезокерамики, поляризованной в направлении оси *Oz*, различные ненулевые модули упругости в массиве `TBDATA` будут иметь следующие номера (для пьезокерамики $c_{22}^E = c_{11}^E$; $c_{23}^E = c_{13}^E$; $c_{55}^E = c_{44}^E$; $c_{66}^E = (c_{11}^E - c_{12}^E)/2$):

№	1	2	3	7	8	12	16	19	21
$c_{\alpha\beta}^E$, пьезокера мика	c_{11}^E	c_{12}^E	c_{13}^E	c_{11}^E	c_{13}^E	c_{33}^E	c_{66}^E	c_{44}^E	c_{44}^E

Для двумерных задач, как видно из сравнения (1.1) – (1.3), достаточно задавать только первые 16 позиций в одномерном массиве модулей $c_{\alpha\beta}^E$.

Заметим, что для плоских и осесимметричных двумерных задач обычно удобно считать в рабочей плоскости *Oxy* ось *Oy* направлением предварительной поляризации пьезокерамики (т.е. осью *z*=(3) для модулей $c_{\alpha\beta}^E$). В этом случае, для плоских и осесимметричных задач различные ненулевые модули упругости в массиве `TBDATA` будут иметь номера:

№	1	2	3	7	8	12	16
$c_{\alpha\beta}^E$, пьезокерамика	c_{11}^E	c_{13}^E	c_{12}^E	c_{33}^E	c_{13}^E	c_{11}^E	c_{44}^E

Аналогичным образом в нестандартном формате задаются в ANSYS и пьезомодули $e_{i\alpha}$. Пьезомодули располагаются в следующем порядке в матрицах размера 6x3 (4x2 – для двумерных задач):

$$\mathbf{e}_{ANSYS,3D} = \begin{array}{c|ccc} & x & y & z \\ \hline x & e_{11} & e_{21} & e_{31} \\ y & e_{12} & e_{22} & e_{32} \\ z & e_{13} & e_{23} & e_{33} \\ xy & e_{16} & e_{26} & e_{36} \\ yz & e_{14} & e_{24} & e_{34} \\ xz & e_{15} & e_{25} & e_{35} \end{array} ; \quad \mathbf{e}_{ANSYS,2D} = \begin{array}{c|cc} & x & y \\ \hline x & e_{11} & e_{21} \\ y & e_{12} & e_{22} \\ z & e_{13} & e_{23} \\ xy & e_{16} & e_{26} \end{array}$$

После команды TB,PIEZ,MAT массив данных TBDATA для пьезомодулей в ANSYS заполняется из массива $\mathbf{e}_{ANSYS,3D}$ по строкам, как одномерный массив размера 18. В результате получается следующее соответствие между данными из массива TBDATA и пьезомодулями $e_{i\alpha}$:

№	1	2	3	4	5	6	7	8	9
$e_{i\alpha}$	e_{11}	e_{21}	e_{31}	e_{12}	e_{22}	e_{32}	e_{13}	e_{23}	e_{33}
№	10	11	12	13	14	15	16	17	18
$e_{i\alpha}$	e_{16}	e_{26}	e_{36}	e_{14}	e_{24}	e_{34}	e_{15}	e_{25}	e_{35}

причем для двумерных задач достаточно задавать элементы с номерами 1,2,4,5,7,8,10 и 11.

Для пьезокерамики, поляризованной в направлении оси Oz , различные ненулевые пьезомодули для трехмерных задач будут иметь следующие номера в массиве TBDATA (для пьезокерамики $e_{32} = e_{31}$; $e_{24} = e_{15}$):

№	3	6	9	14	16
$e_{i\alpha}$, пьезокерамика	e_{31}	e_{31}	e_{33}	e_{15}	e_{15}

Наконец, для плоских и осесимметричных двумерных задач в случаях, когда в плоскости Oxy ось Oy является осью предварительной поляризации пьезокерамики (т.е. осью $z=(3)$ для пьезомодулей $e_{i\alpha}$), различные ненулевые

пьезомодули в массиве TBDATA будут иметь номера, указанные в следующей таблице:

№	2	5	8	10
$e_{i\alpha}$, пьезокерамика	e_{31}	e_{33}	e_{31}	e_{15}

Суммируя изложенное выше, для пьезокерамического материала можно написать следующие фрагменты программ для ANSYS, задающие наборы материальных констант MAT с номером 1 для трехмерных (3D) и двумерных (2D) задач, причем в последнем случае пьезокерамика считается поляризованной в рабочей плоскости Oxy вдоль оси Oy :

Фрагмент 1 (3D).

```
MP, DENS, 1, RHO
TB, ANEL, 1
TBDATA, 1, C11E, C12E, C13E
  TBDATA, 1, C11E, C13E, C12E
TBDATA, 7, C11E, C13E
TBDATA, 12, C33E
TBDATA, 16, C66E
TBDATA, 19, C44E
TBDATA, 21, C44E
TB, PIEZ, 1
TBDATA, 3, E31
TBDATA, 6, E31
TBDATA, 9, E33
TBDATA, 14, E15
TBDATA, 16, E15
MP, PERX, 1, EPS11
MP, PERZ, 1, EPS33
```

Фрагмент 2 (2D).

```
MP, DENS, 1, RHO
TB, ANEL, 1
TBDATA, 7, C33E, C13E
TBDATA, 12, C11E
TBDATA, 16, C44E
TB, PIEZ, 1
TBDATA, 2, E31
TBDATA, 5, E33
TBDATA, 8, E31
TBDATA, 10, E15
MP, PERX, 1, EPS11
MP, PERY, 1, EPS33
```

Здесь $RHO=\rho$; $C11E=c_{11}^E$ и т.п. Эти величины являются скалярными параметрами и должны быть определены раньше. Кроме того, в представленных фрагментах добавлены команды MP, задающие плотность и диэлектрические проницаемости пьезокерамики.

Как видно, задание констант пьезоэлектрических материалов требует достаточных усилий. Важно подчеркнуть, что в представленных фрагментах для определяемых наборов констант оси $Oxy(z)$ являются осями элементных систем координат!

Команда **ET,1,PLANE223,1001,,1** задает четырехузловой конечный элемент PLANE223 с опцией пьезоэлектрического анализа и опцией осесимметричности, который будет использован для решения задачи. Для пользователя все отличие осесимметричной задачи от двумерной задачи о

плоской деформации состоит только в третьей опции данного конечного элемента. (Если записать **ET,1,PLANE223,1001,,0** то будет решаться двумерная задача о плоском напряженном состоянии, если записать **ET,1,PLANE223,1001,,2**, то будет решаться плоской деформации.) Однако ANSYS будет при этой опции использовать уравнения осесимметричной теории пьезоэлектричества, значительно более сложные, чем уравнения плоского напряженного состояния или плоской деформации.

Построение твердотельной модели меридионального сечения пьезоэлектрического диска

При программировании на языке APDL ANSYS твердотельная модель исходной области со сложной геометрией обычно строится «снизу вверх», начиная с построения наиболее простых сущностей (Entities) – опорных точек (Keypoints) и заканчивая построением областей (Areas) для двумерных задач или объемов (Volumes) для трехмерных задач.

Для построения опорных точек используется команда **K**, первым аргументом которой является номер точки, а остальными – координаты точки. Например, команда **K, 2, R1, -H/2** создает точку с номером 2 и координатами $x=R1$, $y=-H/2$, $z=0$ (значение по умолчанию). Для построения линии (сущность Line) между двумя точками используется команда **L**. Например, команда **L, 2, 3** создает линию в текущей системе координат между точками с номерами 2 и 3. При этом создаваемые линии нумеруются в программе автоматически, начиная с наименьшего доступного номера. Для построения областей можно использовать команды **A** и **AL**, при этом создаваемые области также нумеруются автоматически. Команда **AL** строит область по указанным линиям, перечисленным в порядке обхода по или против часовой стрелки (можно перечислить не более 10 линий), а команда **A** аналогично строит область по указанным опорным точкам (можно перечислить не более 18 точек). При этом, если между двумя точками определена линия, то команда **A** будет использовать ее при построении области, в противном случае в текущей координатной системе будет построена недостающая линия.

Согласно рис. 2, в меридиональном сечении диска две зоны с различным направлением поляризации, поэтому твердотельную осесимметричную модель диска можно построить состоящей из двух областей. На рис. 3 показаны построенные области A1 и A2 с нанесенными номерами областей и опорных точек. (Пункты меню Plot->Areas, для нумерации областей и точек PlotCtrls->Numbering->отметить Area numbers, Keypoint numbers)




Рис. 3 Нумерация областей модели меридионального сечения диска




Рис. 4 Линии с заданным разбиением и опорные точки модели меридионального сечения диска

Линии и точки, составляющие области A1 и A2, можно увидеть на рис. 4 (пункты меню Plot->Lines (здесь предварительно в Select->Entities выбраны линии L1-L9), для показа нумерации линий и точек PlotCtrls->Numbering->отметить Line numbers, Keypoint numbers). Здесь прямоугольная область A2 построена обычным образом из четырех линий L3, L4, L5, L9 и четырех точек 3, 4, 5, 6, поскольку верхняя и нижняя границы данной область совпадают с линиями расположения электродов. Область A1 состоит из шести точек 1, 2, 3, 6, 7 и шести линий L1, L2, L9, L7, L7, L8. Для этой области как верхнюю, так и нижнюю границы нужно строить из двух линий, причем опорные точки линий L1 и L7, моделирующих электроды, должны совпадать с концами электродов. Это требуется для того, чтобы при построении конечно-элементной сетки на концы электродов попали узлы конечных элементов. Следует обратить внимание на то, что линия L9 является общей для смежных областей A1 и A2.

Задание направления вектора поляризации для областей пьезоэлектрического диска.

Для плоских и осесимметричных двумерных задач удобно считать ось Oy осью предварительной поляризации в рабочей плоскости Oxy , поэтому в ANSYS для пьезоматериалов с различными векторами поляризации достаточно задавать для соответствующих областей элементные системы координат, повернутые нужным образом относительно глобальной системы координат.




Рис. 5 Направление векторов поляризации для областей меридионального сечения диска

В рассматриваемом примере направление вектора поляризации в области A1 совпадает с направлением оси Oy глобальной декартовой системы координат, поэтому для конечных элементов этой области не требуется вводить специальную элементную систему координат. Для области A2 направление вектора поляризации не совпадает с направлением оси Oy глобальной декартовой системы координат, поэтому для этой области требуется задать элементную систему координат с осью Oy , направленной вдоль вектора поляризации. Для этого элементную систему координат нужно повернуть на 180 градусов относительно глобальной декартовой системы координат. Следующий блок команд создает локальную систему координат и задает ее в качестве элементной для области A2.

```
LOCAL,11,0,,,180
ASEL,S,AREA,,2
AATT,1,1,11
```

На рис. 6 зеленым цветом показаны направления оси Oy элементных систем координат. (Пункты меню Plot->Elements, для показа зон поляризации разным цветом PlotCtrls->Numbering->Elem/Attrib numbering->выбрать Element CS num, Numbering shown with ->Colors only, для показа направления элементной системы координат на каждом элементе PlotCtrls->Symbols, отметить ESYS Element coordinate sys)




Рис. 6 Элементные системы координат в областях меридионального сечения диска: общий вид и увеличенный фрагмент

Построение конечно-элементной сетки с регулярным разбиением.

Поскольку элемент **PLANE223** является четырехугольным конечным элементом и моделируемая область меридионального сечения диска является четырехугольником, а конкретнее, прямоугольником, то логично использовать *регулярное* конечно-элементное разбиение на четырехугольники (даже – на прямоугольники).

В ANSYS существует два типа конечно-элементного разбиения: регулярное (Mapped Mesh) и свободное (Free Mesh). *Свободным* называется

разбиение, которое не имеет ограничений на форму элементов и не имеет заданного шаблона сетки. Обычно свободное разбиение используется для областей и объемов сложной формы. Для свободного разбиения достаточно задать средний размер конечного элемента. В противоположность свободному разбиению *регулярное разбиение* определяется заданной формой конечных элементов (например, только треугольники или только четырехугольники в случае разбиения областей) и заданным шаблоном конечно-элементной сетки с четкими «рядами» элементов. Регулярное разбиение выгодно использовать для областей и объемов простой формы, использование регулярного разбиения помогает значительно сократить время расчетов, и оно, как правило, более точное, чем сравнимое по числу элементов и узлов свободное разбиение.

Для того чтобы построить регулярное разбиение четырехугольной области на четырехугольные конечных элементы, область должна удовлетворять двум критериям: 1) состоять из четырех линий, 2) противоположные стороны области должны быть разбиты на одинаковое число частей. Для области, состоящей из более, чем четырех линий, следует использовать конкатенацию смежных линий, чтобы сократить общее число линий до четырех.

В рассматриваемом примере меридионального сечения диска в принципе возможно два способа построения областей твердотельной модели. Например, можно было построить дополнительную линию можно точками 2 и 7 и составить три прямоугольных области, каждая из которых состояла бы из четырех линий.

Принятый здесь способ позволяет обойтись двумя областями, при этом область A2 является регулярной, а область A1 – нерегулярной, так как состоит из шести линий. Для превращения области A1 в регулярную достаточно объединить между собой линии L1 и L2, которые составляют нижнюю границу области, а также линии L6 и L7, составляющие верхнюю границу области. Для этого служит команда **LCCAT**, которая объединяет две линии в одну для последующего обеспечения регулярного разбиения. Если требуется объединить более двух линий, то в качестве аргумента команды **LCCAT** следует задать **ALL**, при этом будут объединены все линии выбранные ранее с помощью команды выбора линий **LSEL**. Следует помнить, что для построения регулярной конечно-элементной сетки суммарное число делений противоположных линий области должно быть одним и тем же! На рис. 5 показан результат объединения линий верхней и нижней границ области A1. Полученные в результате линии L10 и L11 сохраняют заданное разбиение составляющих их линий (см. рис. 6).




Рис. 7 Конкатенация линий модели меридионального сечения диска

Задание граничных условий.

1) **Граничные условия на электродированных поверхностях.** В теории пьезоэлектричества пьезоэлектрические материалы рассматриваются как активные поляризующиеся диэлектрики. Части их границ Γ_{Vm} ; $m=1,2,\dots,M$; могут быть металлизированными поверхностями или электродами. Электроды Γ_{Vm} считаются эквипотенциальными поверхностями, т.е. на них электрический потенциал $\varphi(\mathbf{x},t)$ не должен зависеть от \mathbf{x} :

$$\varphi = V_m(t); \quad \mathbf{x} \in \Gamma_{Vm} \quad (2.1)$$

Граничные условия (2.1) являются главными или кинематическими электрическими граничными условиями для краевых задач пьезоэлектричества, если $V_m(t)$ – известные функции. Электроды с известными $V_m(t)$ называют электродами, запитываемые генераторами напряжений, а при $V_m = 0$ – короткозамкнутыми заземленными электродами.

Другой тип электродов представляют электроды Γ_{Ql} , запитываемые генераторами токов I_l . Для электродов Γ_{Ql} ; $l=1,2,\dots,L$; граничные условия имеют вид:

$$\varphi = \Phi_l(t); \quad \mathbf{x} \in \Gamma_{Ql} \quad (2.2)$$

$$\int_{\Gamma_{Ql}} \mathbf{n} \cdot [\mathbf{D}] d\Gamma = Q_l; \quad \dot{Q}_l = \pm I_l$$

где \mathbf{n} – нормаль к Γ_{Ql} ; $[\mathbf{D}]$ – скачок вектора электрической индукции \mathbf{D} ; Q_l – суммарный поверхностный заряд на электроде Γ_{Ql} ; причем значения потенциалов $\Phi_l(t)$ изначально не известны. При $Q_l = 0$ электроды (2.2) называются разомкнутыми или пассивными электродами.

Граничные условия (2.1) и (2.1) можно реализовать в ANSYS следующим образом. Узлы КЭ сетки, принадлежащие отдельному электроду Γ_{Vm} или Γ_{Ql} , свяжем в один узел (coupled DOF) командой **CP**. Далее, для выбранного по **CP** представителя – узла с номером N_VOLT можно либо задать значения потенциала $VM=V_m$ по команде **D,N_VOLT,VOLT,VM**, либо определить суммарный заряд $QL=Q_l$ командой **F,N_VOLT,CHRG,QJ** (для КЭ PLANE223, SOLID226, SOLID227) или командой **F,N_VOLT,AMPS,QJ** (для КЭ PLANE13, SOLID5, SOLID98). Блок команд для определения группы узлов

и задания значения потенциала на первом электроде выглядит следующим образом:

```
NSEL,S,LOC,Y,-H/2
NSEL,R,LOC,X,0,R1
CP,1,VOLT,ALL
*GET,N_VOLT1,NODE,,NUM,MIN
D,N_VOLT1,VOLT,-VINP
```

Если требуется задать свободный электрод, то достаточно определить группу командой **CP**, но не использовать команду **D**.

2) «Механические» граничные условия. В рассматриваемом примере нужно поставить условие симметрии относительно оси вращения (оси Oy) и условие закрепления средней крайней точки по оси Oy .

Ограничение на степени свободы (для элемента PLANE223 с опцией пьезоэлектрического анализа их три: UX, UY, VOLT) можно задать либо с помощью команды **D**, которая задает ограничение на степени свободы на узлах, либо с помощью команды **DL**, которая задает ограничение на степени свободы на линиях. Следует иметь в виду, что все «твердотельные» граничные условия будут преобразованы в «конечно-элементные» граничные условия на этапе решения задачи. При этом, «твердотельные» граничные условия на заданной линии имеют приоритет над «конечно-элементными» граничными условиями на узлах той же линии.

2.1) **Условие симметрии** можно задать как с помощью команды **DL**, так и с помощью команды **D**. В команде **DL** можно использовать опцию SYMM (как это сделано в примере: **DL,8,,SYMM**). Другой способ состоит в том, чтобы выбрать нужные узлы и воспользоваться командой **D**. Выбрать узлы, лежащие на линии L8, можно с помощью команд:

```
LSEL,S,LINE,,8
```

```
NSLL,S,1 ! Выбор узлов, принадлежащих выбранным линиям
```

! 2й аргумент – ключ выбора внутренних узлов (здесь 1 – выбор внутренних и внешних узлов)

Команда **D,ALL,UX,0** задает для всех выбранных узлов равенство нулю компоненты вектора перемещений UX, что соответствует симметрии относительно оси Oy . Аналогично команда **D,ALL,UY,0** будет задавать для всех выбранных узлов равенство нулю компоненты вектора перемещений UY, что будет соответствовать симметрии относительно оси Ox .

Отметим, что как можно проверить, в рассматриваемой осесимметричной задаче в ANSYS условия симметрии на линии $x = 0$ будут обеспечиваться автоматически, и команда **DL,8,,SYMM** в примере приведена для надежности.

2.2) **Условие жесткого закрепления** предполагает, что для заданных узлов вектор перемещений равен нулю, т.е. в случае плоской задачи $UX=0$ и $UY=0$. Соответствующий блок команд будет иметь вид

```
D,ALL,UX,0
```

```
D,ALL,UY,0
```

При этом в рассматриваемом примере используется только закрепление по оси Oy , т.е. $UY=0$.

Конечно-элементная модель пьезоэлектрического диска с граничными условиями показана на рис. 8. (Пункты меню Plot->Elements, для отображения граничных условий PltCtrls->Symbols->отметить All applied BC)




Рис. 8 Конечно-элементная сетка с граничными условиями

Просмотр результатов.

В меню ANSYS это можно сделать, используя, например, следующие перемещения по меню:

General Postproc → Plot Results → Deformed Shape (деформированная форма)

General Postproc → Plot Results → Contour Plot → Nodal Solu... → DOF Solution →

X-Component of displacement (для изображения картины распределения перемещений U_x)

Y-Component of displacement (аналогично для изображения картины распределения перемещений U_y)

Displacement vector sum (для вывода модуля вектора перемещений)

Electric potential (для вывода распределения электрического потенциала)

General Postproc → Plot Results → Vector Plot → Predefined

DOF solution → Translation U (для вывода распределения вектора перемещений)

Flux & gradient → Elec field EF (для вывода распределения вектора электрического поля)

Аналогичным образом можно получить распределения различных компонент напряжений, деформаций, вектора напряженности электрического поля и вектора электрической индукции.

Результаты расчетов

В результате расчетов в ANSYS 11.0 по программе, содержащейся в прилагаемом входном файле, в конце будет выполнена команда PLNSOL,U,Y и будет показана картинка распределения перемещений UY . Здесь следует помнить, что решалась осесимметричная задача в меридиональной плоскости,

ось OY соответствует оси Oz в трехмерном пространстве, а ось OX – оси Or . Поэтому UY есть в реальности перемещение u_z , UX – u_r и т.д.

Основные результаты расчетов показаны на рис. 9-14. Распределение механических и электрических характеристик показано на недеформированной форме диска.

```
DISPLACEMENT
STEP=1
SUB =1
TIME=1
DMX =.798E-09
```




Рис. 9 Картина деформированной формы пьезоэлектрического диска

```
NODAL SOLUTION
STEP=1
SUB =1
TIME=1
UX (AVG)
RSYS=0
DMX =.798E-09
SMN =-.757E-09
SMX =.414E-09
```

ANSYS




Рис. 10 Распределение перемещений UX

```
NODAL SOLUTION
STEP=1
SUB =1
TIME=1
UY (AVG)
RSYS=0
DMX =.798E-09
SMN =-.262E-09
SMX =.262E-09
```

ANSYS




Рис. 11 Распределение перемещений UY

VECTOR
STEP=1
SUB =1
TIME=1
U
NODE=1298
MIN=0
MAX=.798E-09

ANSYS




Рис. 12 Распределение вектора перемещений U

NODAL SOLUTION
STEP=1
SUB =1
TIME=1
VOLT (AVG)
RSYS=0
DMX =.798E-09
SMN =-1
SMX =1

ANSYS




Рис. 13 Распределение электрического потенциала ϕ

VECTOR
 STEP=1
 SUB =1
 TIME=1
 EF
 ELEM=352
 MIN=38.2379
 MAX=1433.76

ANSYS




Рис. 14 Распределение вектора электрического поля E

Проанализируем полученные результаты, как с физической точки зрения, так и с математической. Между электродами 1 и 2 (т.е. между сплошными электродами в зоне $0 \leq X \leq R1$) и между электродами 3 и 4 (т.е. между кольцевыми электродами в зоне $R2 \leq X \leq R$) подается разность потенциалов $\Delta V = -2V_{inp}$. Тогда в областях между этими электродами следует ожидать электрическое поле порядка $E \sim \Delta V/H = -2V_{inp}/H = -1000$ (В/м), поскольку $V_{inp} = 1$ (В), $H = 0.002$ (м). Эти значения электрических полей и наблюдаются на рис. 14 (векторы желтого цвета). Для зоны $0 \leq X \leq R1$ между сплошными электродами направление вектора поляризации P противоположно направлению генерируемого поля E . Поэтому в силу явления обратного пьезоэффекта пьезоэлемент в этой области должен сжиматься. Наоборот, в зоне $R2 \leq X \leq R$ между кольцевыми электродами направление вектора поляризации P совпадает с направлением генерируемого поля E . Тогда также в силу явления обратного пьезоэффекта пьезоэлемент в области $R2 \leq X \leq R$ должен расширяться. Именно такие деформации зон под электродами наглядно демонстрирует рис. 9. Таким образом, ожидаемая физика электромеханических процессов в рассматриваемой задаче для пьезокерамического преобразователя с многоэлектродным покрытием подтверждается результатами, представленными на рисунках.

При проведении конечно-элементных расчетов важно помнить, что они являются приближенными. Поэтому необходимо ответить на вопрос, насколько точны полученные результаты, а в каких случаях не стоит даже ожидать сходимости. В приведенном выше тексте программы характеристики конечно-элементной сетки определялись в следующем блоке команд:

! Параметры конечно-элементной сетки
SM=1.00 ! масштабный множитель (scaling multiplier)

HDIV=8*SM ! Кол-во КЭ по толщине диска (должно быть четным числом!)

R1DIV=16*SM ! Кол-во КЭ по радиальному направлению от 0 до R1

R12DIV=32*SM ! Кол-во КЭ по радиальному направлению от R1 до R2

R23DIV=16*SM ! Кол-во КЭ по радиальному направлению от R2 до R

Поэтому для анализа сходимости можно провести расчеты по приведенной программе, меняя параметр SM . Результаты серии таких расчетов приведены в таблице. Как видно из таблицы, максимальные значения перемещений стабилизируются уже при $SM=0.75$ и далее при увеличении параметра SM , т.е. при уменьшении размеров КЭ, значения перемещений практически не меняются. Таким образом, из данных таблицы можно сделать вывод о том, что для оценки максимальных смещений в данной задаче достаточно взять $SM=0.75$, но и при $SM=0.25$ относительная погрешность определения перемещений оказывается менее 3 %, что вполне достаточно для большинства практических приложений. Значение $SM=0.25$ означает, что по толщине диска берется всего два конечных элемента ($HDIV = 2$). Такое достаточно грубое разбиение здесь оказывается удовлетворительным для определения смещений, поскольку рассматриваемая задача статическая, имеет достаточно простую геометрию, а элементы PLANE223 являются КЭ с квадратичной аппроксимацией по каждой канонической переменной.

Таблица. Данные для анализа сходимости результатов

SM	0.25	0.50	0.75	1.00	1.25	1.50	1.75	2.00
$\max u_y \cdot 10^{-12}$ (м)	255	260	262	262	262	262	262	262
$\max E $ (В/м)	1001	1141	262	1293	1564	1685	1798	1904

Между тем, как видно из рис. 14, вектор напряженности электрического поля достаточно сильно изменяется в окрестности границ электродированных поверхностей (векторы кранного цвета). Данные из таблицы показывают, что с увеличением плотности КЭ сетки максимальные значения модуля вектора напряженности электрического поля возрастают, и никакой сходимости этого значения не наблюдается. Этот эффект также вполне ожидаем, поскольку в задачах электроупругости поверхностные электроды являются концентраторами для различных компонент электромеханических полей напряжений, деформаций, напряженности электрического поля и электрической индукции (т.е. для градиентных или потоковых величин, определяемых через производные от перемещений и электрического потенциала). Роль поверхностных электродов в таких задачах аналогична роли штампов с негладкими границами в контактных задачах. Поэтому, как и в аналогичных контактных задачах, вблизи границ электродов некоторые компоненты напряжений, деформаций, напряженности электрического поля и электрической индукции могут иметь корневые особенности типа $r^{1/2}$, где r – расстояние от границы электрода. Естественно, что тогда при использовании

обычных изопараметрических КЭ будет наблюдаться рост величин этих полей вблизи границ электродов. Для практики это означает, что конкретные значения подобных потоковых величин в малой области вблизи границ электродов здесь не имеют особого смысла, но вдали от границ электродов сходимость будет иметь место. Также будут сходиться и их интегральные величины (например, суммарный электрический заряд).

Сказанное выше иллюстрирует график (рис. 15) изменения осевой компоненты напряженности электрического поля E_z (EF, Y – в ANSYS) вдоль радиуса диска на его верхней поверхности ($0 \leq X \leq R$, $Y = H/2$ – в ANSYS), построенный с помощью следующих команд постпроцессора ANSYS:

```

! Постпроцессорный вывод графиков
/PLOPTS,INFO,ON
/COLOR,CURVE,WHIT,1,6 ! цвет графиков - белый (черный при
инвертации)
/COLOR,GRID,WHIT, ! цвет сетки - белый (черный при инвертации)
/PLOPTS,TITLE,OFF ! заголовка графика нет
/PLOPT,FRAME,OFF !рамки нет
/AXLAB,X,r !подпись к оси X (r)
/AXLAB,Y, E_Z ! подпись к оси Y (z)
/GROPT,DIVX,4 ! Количество делений по оси X
/GROPT,DIVY,4 ! Количество делений по оси Y
/GROPT,DIG2,1 ! Количество цифр после запятой
! Команды определения пути с опорными точками
PATH,XX,4,,120
PPATH,1,,0,H/2
PPATH,2,,R1,H/2
PPATH,3,,R2,H/2
PPATH,4,,R,H/2 ! Определение величины E_Z для построения графика
вдоль пути
PDEF,E_Z,EF,Y ! Определение других величин (если понадобится для
них строить графики)
PDEF,T_ZZ,S,Y $ PDEF,T_RR,S,X
PDEF,D_Z,D,Y $ PDEF,D_R,D,X $ PDEF,E_R,EF,X
PLPATH,E_Z ! Построение графика E_Z вдоль пути

```




Рис. 15 Поведение компоненты E_z вектора электрического поля на поверхности диска

Представленные на рис. 15 конечно-элементные результаты для зависимости $E_z = E_z(r)$ не точны в окрестности границ электродов $R_1 = 0.5 \cdot 10^{-2}$ (м) и $R_2 = 1.5 \cdot 10^{-2}$ (м). Если, например, при $r < R_1$ эта кривая несколько напоминает функцию с корневой особенностью $k\sqrt{R_1^2 - r^2}$, то всплеск функции при $r > R_1$ полностью определяется только погрешностью конечно-элементных аппроксимаций и принятых способов осреднения при вычислении градиентов полей. Однако вдали от границ электродов поле определяется достаточно точно.

Аналогичным образом можно проанализировать и поведение других градиентных величин электромеханических полей: E_r , D_z , D_r , σ_{zz} , и т.д.